
Subject: Re: How to read data with format codes?
Posted by [penteado](#) on Sun, 14 Mar 2010 01:19:50 GMT
[View Forum Message](#) <> [Reply to Message](#)

The easiest way I see is not with format codes, but with `read_ascii` with a template:

```
templ=ascii_template(file)
a=read_ascii(file,template=templ)
```

If you want it to be non-interactive, you can either use `ascii_template` once, then save the resulting template, or generate one yourself, as in:

```
template={version:1.0,datastart:0L,delimiter:byte('|'),missingvalue:!  
values.d_nan,$  
commentsymbol:',fieldcount:[4L],fieldtypes:[0L,5L,5L,5L],$  
fieldnames:['FIELD1','FIELD2','FIELD3','FIELD4'],$  
fieldlocations:[0L,0L,0L,0L],fieldgroups:[0L,1L,2L,3L]}
```

The most relevant here being `delimiter`, `fieldcount` and `fieldtype`.

If you had commas instead of "|" separating the columns, none of that would be needed, you would be able to just use `read_csv`.

Another way would be to read the file into a string array, then use `strsplit` to separate each line at the "|":

```
nlines=file_lines(file)
a=strarr(nlines)
openr,unit,file,/get_lun
readf,unit,a
free_lun,unit
a=strtrim(a,2)
b=dblarr(3,nlines)
for i=0,nlines-1 do b[*,i]=strsplit(a[i],'|',/extract)
```
